

# NAG Library Routine Document

## F07PUF (ZHPCON)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07PUF (ZHPCON) estimates the condition number of a complex Hermitian indefinite matrix  $A$ , where  $A$  has been factorized by F07PRF (ZHPTRF), using packed storage.

### 2 Specification

SUBROUTINE F07PUF (UPLO, N, AP, IPIV, ANORM, RCOND, WORK, INFO)

INTEGER N, IPIV(\*), INFO  
 REAL (KIND=nag\_wp) ANORM, RCOND  
 COMPLEX (KIND=nag\_wp) AP(\*), WORK(2\*N)  
 CHARACTER(1) UPLO

The routine may be called by its LAPACK name *zhpcon*.

### 3 Description

F07PUF (ZHPCON) estimates the condition number (in the 1-norm) of a complex Hermitian indefinite matrix  $A$ :

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1.$$

Since  $A$  is Hermitian,  $\kappa_1(A) = \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty$ .

Because  $\kappa_1(A)$  is infinite if  $A$  is singular, the routine actually returns an estimate of the **reciprocal** of  $\kappa_1(A)$ .

The routine should be preceded by a call to F06UDF to compute  $\|A\|_1$  and a call to F07PRF (ZHPTRF) to compute the Bunch–Kaufman factorization of  $A$ . The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate  $\|A^{-1}\|_1$ .

### 4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

### 5 Parameters

1: UPLO – CHARACTER(1) *Input*

*On entry:* specifies how  $A$  has been factorized.

UPLO = 'U'

$A = PUDU^H P^T$ , where  $U$  is upper triangular.

UPLO = 'L'

$A = PLDL^H P^T$ , where  $L$  is lower triangular.

*Constraint:* UPLO = 'U' or 'L'.

- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: AP(\*) – COMPLEX (KIND=nag\_wp) array *Input*  
**Note:** the dimension of the array AP must be at least  $\max(1, N \times (N + 1)/2)$ .  
*On entry:* the factorization of  $A$  stored in packed form, as returned by F07PRF (ZHPTRF).
- 4: IPIV(\*) – INTEGER array *Input*  
**Note:** the dimension of the array IPIV must be at least  $\max(1, N)$ .  
*On entry:* details of the interchanges and the block structure of  $D$ , as returned by F07PRF (ZHPTRF).
- 5: ANORM – REAL (KIND=nag\_wp) *Input*  
*On entry:* the 1-norm of the **original** matrix  $A$ , which may be computed by calling F06UDF with its parameter NORM = '1'. ANORM must be computed either **before** calling F07PRF (ZHPTRF) or else from a **copy** of the original matrix  $A$ .  
*Constraint:* ANORM  $\geq 0.0$ .
- 6: RCOND – REAL (KIND=nag\_wp) *Output*  
*On exit:* an estimate of the reciprocal of the condition number of  $A$ . RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than **machine precision**,  $A$  is singular to working precision.
- 7: WORK( $2 \times N$ ) – COMPLEX (KIND=nag\_wp) array *Workspace*
- 8: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed estimate RCOND is never less than the true value  $\rho$ , and in practice is nearly always less than  $10\rho$ , although examples can be constructed where RCOND is much larger.

## 8 Further Comments

A call to F07PUF (ZHPCON) involves solving a number of systems of linear equations of the form  $Ax = b$ ; the number is usually 5 and never more than 11. Each solution involves approximately  $8n^2$  real floating point operations but takes considerably longer than a call to F07PSF (ZHPTRS) with one right-hand side, because extra care is taken to avoid overflow when  $A$  is approximately singular.

The real analogue of this routine is F07PGF (DSPCON).

## 9 Example

This example estimates the condition number in the 1-norm (or  $\infty$ -norm) of the matrix  $A$ , where

$$A = \begin{pmatrix} -1.36 + 0.00i & 1.58 + 0.90i & 2.21 - 0.21i & 3.91 + 1.50i \\ 1.58 - 0.90i & -8.87 + 0.00i & -1.84 - 0.03i & -1.78 + 1.18i \\ 2.21 + 0.21i & -1.84 + 0.03i & -4.63 + 0.00i & 0.11 + 0.11i \\ 3.91 - 1.50i & -1.78 - 1.18i & 0.11 - 0.11i & -1.84 + 0.00i \end{pmatrix}.$$

Here  $A$  is Hermitian indefinite, stored in packed form, and must first be factorized by F07PRF (ZHPTRF). The true condition number in the 1-norm is 9.10.

### 9.1 Program Text

```

Program f07pufe

!      F07PUF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x02ajf, zhpcon, zhptraf, zlanhp => f06udf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: anorm, rcond
      Integer                     :: i, info, j, n
      Character (1)               :: uplo
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ap(:), work(:)
      Real (Kind=nag_wp), Allocatable  :: rwork(:)
      Integer, Allocatable          :: ipiv(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07PUF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (ap(n*(n+1)/2),work(2*n),rwork(n),ipiv(n))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
         Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
      Else If (uplo=='L') Then
         Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
      End If

!      Compute norm of A
!      f06udf is the NAG name equivalent of the LAPACK auxiliary zlanhp
      anorm = zlanhp('1-norm',uplo,n,ap,rwork)

!      Factorize A
!      The NAG name equivalent of zhptraf is f07prf
      Call zhptraf(uplo,n,ap,ipiv,info)

      Write (nout,*)
      If (info==0) Then

!      Estimate condition number
!      The NAG name equivalent of zhpcon is f07puf
      Call zhpcon(uplo,n,ap,ipiv,anorm,rcond,work,info)

      If (rcond>=x02ajf()) Then
         Write (nout,99999) 'Estimate of condition number =', &
            1.0_nag_wp/rcond
      End If

```

```

      Else
        Write (nout,*) 'A is singular to working precision'
      End If
    Else
      Write (nout,*) 'The factor D is singular'
    End If

99999 Format (1X,A,1P,E10.2)
      End Program f07pufe

```

## 9.2 Program Data

F07PUF Example Program Data

```

4                                     :Value of N
'L'                                   :Value of UPLO
(-1.36, 0.00)
( 1.58,-0.90) (-8.87, 0.00)
( 2.21, 0.21) (-1.84, 0.03) (-4.63, 0.00)
( 3.91,-1.50) (-1.78,-1.18) ( 0.11,-0.11) (-1.84, 0.00) :End of matrix A

```

## 9.3 Program Results

F07PUF Example Program Results

Estimate of condition number = 6.68E+00

---